```
;;;;;;; I2C subroutines ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;
; The I2Cout subroutine transfers out three bytes: DEVADD, INTADD, and DATAOUT

I2Cout
        call    Start           ;Generate START condition
        movf    DEVADD,W        ;Send peripheral address with R/W=0 (write)
        call    TX
        movf    INTADD,W        ;Send peripheral's internal address
        call    TX
        movf    DATAOUT,W       ;Send data to write to peripheral
        call    TX
        call    Stop            ;Generate STOP condition
        return

; The I2Cin subroutine transfers out DEVADD (with R/W=0) and INTADD, restarts,
; transfers out DEVADD (with R/W=1) and reads one byte back into DATAIN.

I2Cin
        call    Start           ;Generate START condition
        movf    DEVADD,W        ;Send peripheral address with R/W=0 (write)
        call    TX
        movf    INTADD,W        ;Send peripheral's internal address
        call    TX
        call    ReStart         ;ReSTART
        movf    DEVADD,W        ;Send peripheral address
        iorlw   B'00000001'     ; with R/W=1 (read)
        call    TX
        bsf     TXBUFF,7        ;NOACK the following read of one byte
        call    RX              ;Read byte
        movwf   DATAIN          ; into DATAIN
        call    Stop            ;Generate STOP condition
        return

; The Start subroutine initializes the I2C bus and then generates the START
; condition on the I2C bus.
; The ReStart entry point bypasses the initialization of the I2C bus.

Start
        movlw   B'00111011'     ;Enable I2C master mode
        movwf   SSPCON
        bcf     PORTC,SDA       ;Drive SDA low when it is an output
        bcf     PORTC,SCL       ;Drive SCL low when it is an output
        movlw   TRISC           ;Set indirect pointer to TRISC
        movwf   FSR
ReStart
        bsf     INDF,SDA        ;Make sure SDA is high
        bsf     INDF,SCL        ;Make sure SCL is high
        delay   0,1,2           ;t:START
        bcf     INDF,SDA
        delay   0,1,2           ;t:START
        bcf     INDF,SCL
        return

; The Stop subroutine generates the STOP condition on the I2C bus.

Stop
        bcf     INDF,SDA        ;Return SDA low
        bsf     INDF,SCL        ;Drive SCL high
        delay   0,1,2           ;t:STOP
        bsf     INDF,SDA        ; and then drive SDA high
        return
```

Figure 9-9 I²C subroutines.